Substitution without copy and paste

Thorsten Altenkirch
University of Nottingham, Nottingham, United Kingdom

Nathaniel Burke
Imperial College London, London, United Kingdom

Philip Wadler
University of Edinburgh, Edinburgh, United Kingdom

Abstract

When defining substitution recursively for a language with binders like the simply typed λ-calculus, we need to define substitution and renaming separately. When we want to verify the categorical properties of this calculus, we end up repeating the same argument many times. In this paper we present a lightweight method that avoids this repetition and is implemented in Agda.

We use our setup to also show that the recursive definition of substitution gives rise to a simply typed category with families (CwF) and indeed that it is isomorphic to the initial simply typed CwF.

2012 ACM Subject Classification Theory of computation → Type theory

Keywords and phrases Substitution, Metatheory, Agda

Digital Object Identifier 10.4230/LIPIcs...

1 Introduction

> Some half dozen persons have written technically on combinatory logic, and most of these, including ourselves, have published something erroneous. [9]

The first author was writing an introduction to category theory for functional programmers. One example category is that of simply-typed λ-terms and substitutions; and proving the expected category laws seemed a suitable exercise. However, the answer was more difficult than expected, so we attempted to formalise the solution in Agda. The main setback was that the same proofs were repeated many times. One guideline of good software engineering is to **not write code by copy and paste**, and this applies doubly to formal proofs.

This paper is the result of our effort to refactor the proof. The method used also applies to other problems; in particular, we see the current construction as a warmup for the recursive definition of substitution for dependent type theory. This in turn may have interesting applications for coherence, i.e., interpreting dependent types in higher categories.

1.1 In a nutshell

When working with substitution for a calculus with binders, we find that you have to differentiate between renamings ($\Delta \models\mathsf{v} \Gamma$) where variables are substituted only for variables ($\Gamma \ni A$) and proper substitutions ($\Delta \models \Gamma$) where variables are replaced with terms ($\Gamma \vdash A$). This results in having to define several similar operations

$$\_\mathsf{v}[\_]\mathsf{v} : \Gamma \ni A \to \Delta \models\mathsf{v} \Gamma \to \Delta \ni A \qquad \_[\_]\mathsf{v} : \Gamma \vdash A \to \Delta \models\mathsf{v} \Gamma \to \Delta \vdash A$$
$$\_\mathsf{v}[\_] : \Gamma \ni A \to \Delta \models \Gamma \to \Delta \vdash A \qquad \_[\_] : \Gamma \vdash A \to \Delta \models \Gamma \to \Delta \vdash A$$

And indeed the operations on terms depend on the operations on variables. This duplication gets worse when we prove properties of substitution, such as the functor law, $x\,[\,xs \circ ys\,] \equiv x\,[\,xs\,]\,[\,ys\,]$. Since all components x, xs, ys can be either variables/renamings or terms/substitutions, we seemingly need to prove eight possibilities (with the repetition

extending also to the intermediary lemmas). Our solution is to introduce a type of sorts with $\mathsf{V} : \mathsf{Sort}$ for variables/renamings and $\mathsf{T} : \mathsf{Sort}$ for terms/substitutions, leading to a single substitution operation $\_[\_] : \Gamma \vdash [\ \mathsf{q}\ ]\ \mathsf{A} \to \Delta \models [\ \mathsf{r}\ ]\ \Gamma \to \Delta \vdash [\ \mathsf{q} \sqcup \mathsf{r}\ ]\ \mathsf{A}$ where $\mathsf{q}, \mathsf{r} : \mathsf{Sort}$ and $\mathsf{q} \sqcup \mathsf{r}$ is the least upper bound in the lattice of sorts ($\mathsf{V} \sqsubseteq \mathsf{T}$). With this, we only need to prove one variant of the functor law, relying on the fact that $\_ \sqcup \_$ is associative. We manage to convince Agda's termination checker that V is structurally smaller than T (see section 3) and, indeed, our highly mutually recursive definitions relying on this are accepted by Agda.

We also relate the recursive definition of substitution to a specification using a quotient-inductive-inductive type, or QIIT (a mutual inductive type with equations) where substitution is a term former (i.e. explicit substitutions). Specifically, our specification is such that the substitution laws correspond to the equations of a simply typed category with families (CwF) (a variant of a category with families where the types do not depend on a context). We show that our recursive definition of substitution leads to a simply typed CwF which is isomorphic to the specified initial one. This can be viewed as a normalisation result where the usual λ-terms without explicit substitutions are the *substitution normal forms*.

1.2 Related work

[10] introduces his eponymous indices and also the notion of simultaneous substitution. We are here using a typed version of de Bruijn indices, e.g. see [6] where the problem of showing termination of a simple definition of substitution (for the untyped λ-calculus) is addressed using a well-founded recursion. The present approach seems to be simpler and scales better, avoiding well-founded recursion. Andreas Abel used a very similar technique to ours in his unpublished Agda proof [1] for untyped λ-terms when implementing [6].

The monadic approach has been investigated in [14], where the duplication between renamings and substitutions is factored into *kits*. The structure of the proofs is explained in [3] from a monadic perspective. Indeed this example is one of the motivations for relative monads [7]. In the monadic approach, we represent substitutions as functions;, however it is not clear how to extend this to dependent types without "very dependent" [11, 5] types.

There are a number of publications on formalising substitution laws. Just to mention a few recent ones: [17] develops a Rocq library which automatically derives substitution lemmas, but the proofs are repeated for renamings and substitutions. Their equational theory is similar to the simply typed CwFs we are using in section 5. [15] uses Agda, but extrinsically (i.e. separating preterms and typed syntax). Here, the approach from [3] is applied to factor the construction using kits. [16] instead uses intrinsic syntax, but with renamings and substitutions defined separately, and relevant substitution lemmas repeated for all required combinations.

1.3 Using Agda

For the technical details of Agda we refer to the online documentation [18]. We generally stick to plain Agda: inductive definitions and structurally recursive programs/proofs. Termination is checked by Agda's termination checker [2] which uses a lexical combination of structural descent that is inferred by the termination checker by investigating all possible recursive paths. We will define mutually recursive proofs which heavily rely on each other.

The only recent feature we use, albeit sparingly, is the possibility to turn propositional equations into rewriting rules (i.e. definitional equalities). This makes the statement of some theorems more readable (avoiding manual transports with subst), but it is not essential.

We extensively use variable declarations to introduce implicit quantification (we summarize the variable conventions in passing in the text). We also use ∀-prefix so we can elide types of function parameters where they can be inferred, i.e. instead of $\{\Gamma : \mathsf{Con}\} \to \ldots$ we just write $\forall\,\{\Gamma\} \to \ldots$ Implicit variables, which are indicated by using $\{\ldots\}$ instead of $(\ldots)$ in dependent function types, can be instantiated using the syntax $\mathsf{f}\,\{\mathsf{x} = \mathsf{y}\}$.

Agda syntax is very flexible, allowing mixfix syntax declarations using '_'s to indicate where the parameters go. In the proofs, we use the Agda standard library's definitions for equational derivations, which exploit this flexibility.

The source of this document contains the actual Agda code, i.e. it is a literate Agda file. Different chapters are in different modules to avoid name clashes, e.g. preliminary definitions from section 2 are redefined later.

2 The naive approach

Let us first review the naive approach which leads to the copy-and-paste proof. We define types (A, B, C) and contexts (Γ, Δ, Θ):

```
data Ty : Set where
  o       : Ty
  _⇒_     : Ty → Ty → Ty

data Con : Set where
  •       : Con
  _▷_     : Con → Ty → Con
```

Next we introduce intrinsically typed de Bruijn variables (i, j, k) and λ-terms (t, u, v):

```
data _∋_ : Con → Ty → Set where
  zero : Γ ▷ A ∋ A
  suc  : Γ ∋ A → (B : Ty) → Γ ▷ B ∋ A

data _⊢_ : Con → Ty → Set where
  `_   : Γ ∋ A → Γ ⊢ A
  _·_  : Γ ⊢ A ⇒ B → Γ ⊢ A → Γ ⊢ B
  λ_   : Γ ▷ A ⊢ B → Γ ⊢ A ⇒ B
```

Here the constructor `_ embeds variables in λ-terms. We write applications as t · u. Since we use de Bruijn variables, lambda abstraction λ_ doesn't bind a name explicitly (instead, variables count the number of binders between them and their actual binding site). We also define substitutions as sequences of terms:

```
data _⊨_ : Con → Con → Set where
  ε   : Γ ⊨ •
  _,_ : Γ ⊨ Δ → Γ ⊢ A → Γ ⊨ Δ ▷ A
```

Now to define the categorical structure (_∘_, id) we first need to define substitution for terms and variables:

```
_v[_] : Γ ∋ A → Δ ⊨ Γ → Δ ⊢ A
zero    v[ ts , t ] = t
(suc i) v[ ts , t ] = i v[ ts ]

_[_] : Γ ⊢ A → Δ ⊨ Γ → Δ ⊢ A
(` i)   [ ts ] = i v[ ts ]
(t · u) [ ts ] = (t [ ts ]) · (u [ ts ])
(λ t)   [ ts ] = λ ?
```

As usual, we encounter a problem with the case for binders λ_. We are given a substitution $\mathsf{ts} : \Delta \models \Gamma$ but the body t lives in the extended context $\mathsf{t} : \Gamma\,,\,\mathsf{A} \vdash \mathsf{B}$. We need to exploit the fact that context extension _▷_ is functorial, $\_\uparrow\_ : \Gamma \models \Delta \to (\mathsf{A} : \mathsf{Ty}) \to \Gamma \rhd \mathsf{A} \models \Delta \rhd \mathsf{A}$. Using _↑_ we can define $(\lambda\,\mathsf{t})\,[\,\mathsf{ts}\,] = \lambda\,(\mathsf{t}\,[\,\mathsf{ts} \uparrow \_\,])$.

However, now we have to define _↑_. This is easy (isn't it?) but we need weakening on substitutions:

```
ts ↑ A = ts ⁺ A , ` zero

_⁺_ : Γ ⊨ Δ → (A : Ty) → Γ ▷ A ⊨ Δ
ε       ⁺ A = ε
(ts , t) ⁺ A = ts ⁺ A , suc-tm t A

suc-tm : Γ ⊢ B → (A : Ty) → Γ ▷ A ⊢ B
```

Now we need to define _⁺_, which is nothing but a fold of weakening of terms:

But how can we define suc-tm when we only have weakening for variables? If we already had

identity id : Γ ⊨ Γ and substitution we could write: suc-tm t A = t [id ⁺ A], but this is certainly not structurally recursive (and hence is rejected by Agda's termination checker).

To fix this, we use the fact that id is a renaming, i.e. it is a substitution only containing variables, and we can easily define _⁺v_ for renamings. This leads to a structurally recursive definition, but we have to repeat the substitution definition for renamings.

```
data _⊨v_ : Con → Con → Set where
  ε : Γ ⊨v •
  _,_ : Γ ⊨v Δ → Γ ∋ A → Γ ⊨v Δ ▷ A
```

```
_v[_]v : Γ ∋ A → Δ ⊨v Γ → Δ ∋ A
zero    v[ is , i ]v = i
(suc i _) v[ is , j ]v = i v[ is ]v

_⁺v_ : Γ ⊨v Δ → ∀ A → Γ ▷ A ⊨v Δ
ε       ⁺v A = ε
(is , i) ⁺v A = is ⁺v A , suc i

_↑v_ : Γ ⊨v Δ → ∀ A → Γ ▷ A ⊨v Δ ▷ A
is ↑v A = is ⁺v A , zero
```

```
_[_]v : Γ ⊢ A → Δ ⊨v Γ → Δ ⊢ A
(` i) [ is ]v = ` (i v[ is ]v)
(t · u) [ is ]v = (t [ is ]v) · (u [ is ]v)
(λ t) [ is ]v = λ (t [ is ↑v _ ]v)

idv : Γ ⊨v Γ
idv {Γ = •}     = ε
idv {Γ = Γ ▷ A} = idv ↑v A

suc-tm t A = t [ idv ⁺v A ]v
```

This may not seem too bad: to ensure structural termination we just have to duplicate a few definitions, but it gets much worse when proving the laws. For example, to prove associativity, we first need to prove functoriality of substitution: [∘] : t [us ∘ vs] ≡ t [us] [vs]. Since t, us, vs can be variables/renamings or terms/substitutions, there are in principle eight combinations. Each time, we must to prove a number of lemmas again in a different setting.

In the rest of the paper we describe a technique for factoring these definitions and the proofs, only relying on the Agda termination checker to validate that the recursion is structurally terminating.

3 Factorising with sorts

Our main idea is to turn the distinction between variables and terms into a parameter. The first approximation of this idea is to define a type Sort (q, r, s):

```
data Sort : Set where
  V T : Sort
```

but this is not exactly what we want; ideally, Agda should know that the sort of variables V is *smaller* than the sort of terms T (following intuition that variable weakening is trivial, but to weaken a term we must construct a renaming). Agda's termination checker only knows about the structural orderings, but with the following definition, we can make V structurally smaller than T>V V isV, while maintaining that Sort has only two elements.

```
data Sort : Set where
  V     : Sort
  T>V : (s : Sort) → IsV s → Sort

data IsV : Sort → Set where
  isV : IsV V
```

Here the predicate isV only holds for V. This particular encoding makes use of Agda's support for inductive-inductive datatypes (IITs), but a pair of a natural number n and a proof n ⩽ 1 would also work, i.e. Sort = Σ ℕ (_ ⩽ 1).

We can now define T = T>V V isV : Sort but, even better, we can tell Agda that this is a derived pattern with pattern T = T>V V isV. This means we can pattern match over Sort just with V and T, while ensuring V is visibly (to Agda's termination checker) structurally smaller than T.

We can now define terms and variables in one go (x, y, z):

```
data _⊢[_]_ : Con → Sort → Ty → Set where
  zero : Γ ▷ A ⊢[ V ] A
  suc  : Γ ⊢[ V ] A → (B : Ty) → Γ ▷ B ⊢[ V ] A
  `_   : Γ ⊢[ V ] A → Γ ⊢[ T ] A
  _·_  : Γ ⊢[ T ] A ⇒ B → Γ ⊢[ T ] A → Γ ⊢[ T ] B
  λ_   : Γ ▷ A ⊢[ T ] B → Γ ⊢[ T ] A ⇒ B
```

While almost identical to the previous definition ($\Gamma \vdash[\,V\,] A$ corresponds to $\Gamma \ni A$ and $\Gamma \vdash[\,T\,] A$ to $\Gamma \vdash A$) we can now parametrize all definitions and theorems explicitly. As a first step, we can generalize renamings and substitutions (xs, ys, zs):

```
data _⊨[_]_ : Con → Sort → Con → Set where
  ε   : Γ ⊨[ q ] •
  _,_ : Γ ⊨[ q ] Δ → Γ ⊢[ q ] A → Γ ⊨[ q ] Δ ▷ A
```

To account for the non-uniform behaviour of substitution and composition (the result is V only if both inputs are V) we define a least upper bound on Sort. We also need this order as a relation.

```
_⊔_ : Sort → Sort → Sort
V ⊔ r = r
T ⊔ r = T
```

```
data _⊑_ : Sort → Sort → Set where
  rfl : s ⊑ s
  v⊑t : V ⊑ T
```

This is just boolean algebra. We need a number of laws:

```
⊑t : s ⊑ T
v⊑ : V ⊑ s
⊑q⊔ : q ⊑ (q ⊔ r)
⊑⊔r : r ⊑ (q ⊔ r)
⊔⊔ : q ⊔ (r ⊔ s) ≡ (q ⊔ r) ⊔ s
⊔v : q ⊔ V ≡ q
```

which are easy to prove by case analysis, e.g. $\sqsubseteq t\,\{V\} = v\sqsubseteq t$ and $\sqsubseteq t\,\{T\} =$ rfl. To improve readability we turn the equations (⊔⊔, ⊔v) into rewrite rules. This introduces new definitional equalities, i.e. $q \sqcup (r \sqcup s) = (q \sqcup r) \sqcup s$ and $q \sqcup V = q$ are now used by the type checker[1].

The order on sorts gives rise to a functor, witnessed by $\mathsf{tm}\sqsubseteq : q \sqsubseteq s \to \Gamma \vdash[\,q\,] A \to \Gamma \vdash[\,s\,] A$, where $\mathsf{tm}\sqsubseteq$ rfl x = x and $\mathsf{tm}\sqsubseteq\ v\sqsubseteq t\ i = `\ i$.

By making functoriality of context extension parameteric, $\_\uparrow\_ : \Gamma \models[\,q\,] \Delta \to \forall\, A \to \Gamma \triangleright A \models[\,q\,] \Delta \triangleright A$, we are ready to define substitution and renaming in one operation:

```
_[_] : Γ ⊢[ q ] A → Δ ⊨[ r ] Γ → Δ ⊢[ q ⊔ r ] A
zero    [ xs , x ] = x
(suc i _) [ xs , x ] = i [ xs ]
```

```
(` i)   [ xs ] = tm⊑ ⊑t (i [ xs ])
(t · u) [ xs ] = (t [ xs ]) · (u [ xs ])
(λ t)   [ xs ] = λ (t [ xs ↑ _ ])
```

We use _⊔_ here to take care of the fact that substitution will only return a variable if both inputs are variables / renamings. We need to use $\mathsf{tm}\sqsubseteq$ to take care of the two cases when substituting for a variable.

We can also implement id using _↑_ (by folding contexts), but to define _↑_ itself, we need parametric versions of zero and suc. zero is easy:

```
id : Γ ⊨[ V ] Γ
id {Γ = •}     = ε
id {Γ = Γ ▷ A} = id ↑ A
```

```
zero[_] : ∀ q → Γ ▷ A ⊢[ q ] A
zero[ V ] = zero
zero[ T ] = ` zero
```

[1] Effectively, this feature allows a selective use of extensional Type Theory.

However, suc is more subtle since the case for T depends on its fold over substitutions:

```
suc[_] :  ∀ q → Γ ⊢[ q ] B → ∀ A                _⁺_ :  Γ ⊨[ q ] Δ → ∀ A
       → Γ ▷ A ⊢[ q ] B                                → Γ ▷ A ⊨[ q ] Δ
suc[ V ] i A = suc i A                          ε    ⁺ A = ε
suc[ T ] t A = t [ id ⁺ A ]                     (xs , x) ⁺ A = xs ⁺ A , suc[ _ ] x A
```

And finally we can define xs ↑ A = xs ⁺ A , zero[_].

3.1 Termination

Unfortunately (as of Agda 2.7.0.1[2]), we now hit a termination error.

The cause turns out to be id. Termination here hinges on weakening for terms (suc[T] t A) building and applying a renaming (i.e. a sequence of variables, for which weakening is trivial) rather than a full substutution. Note that if id produced 'Γ ⊨[T] Γ's, or if we implemented weakening for variables (suc[V] i A) with i [id ⁺ A], our operations would still be type-correct, but would genuinely loop, so perhaps Agda is right to be careful.

Of course, we have specialised weakening for variables, so we now must ask why Agda still doesn't accept our program. The limitation is ultimately a technical one: Agda only looks at the direct arguments to function calls when building the call graph from which it identifies termination order [2]. Because id is not passed a sort, the sort cannot be considered as decreasing in the case of term weakening (suc[T] t A).

Luckily, there is an easy solution here: making id Sort-polymorphic and instantiating with V at the call-sites adds new rows/columns (corresponding to the Sort argument) to the call matrices involving id, enabling the decrease to be tracked and termination to be correctly inferred by Agda. We present the call graph diagramatically (inlining _ ↑ _), in the style of [13].



Function	Measure
$t1^{q1}_{\Gamma 1}\ [\ {}^{r1}\sigma 1^{\Delta 1}_{\Gamma 1}\ ]$	$(r_1\ ,\ t_1)$
$id^{r2}_{\Gamma 2}$	$(r_2\ ,\ \Gamma_2)$
${}^{r3}\sigma 3^{\Delta 3}_{\Gamma 3}\ ⁺\ A$	$(r_3\ ,\ \sigma_3)$
$suc[\ q_4\ ]\ t4^{q4}_{\Gamma 4}$	(q_4)

Table 1 Per-function termination measures

Figure 1 Call graph of substitution operations

To justify termination, we note that along all cycles in the graph, either the Sort strictly decreases in size, or the size of the Sort is preserved and some other argument (the context, substitution or term) gets smaller. Following this, we can assign lexicographically-decreasing measures to each of the functions (Table 1).

In practice, we will generally require identity renamings, rather than substitutions, and so we shall continue as if the original id definition worked (recovering V-only id from the Sort-polymorphic one is easy after all: we merely need to instantiate {q = V}).

[2] We have submitted a PR (#7695) to Agda that extends the termination checking algorithm such that these definitions are accepted directly, so this might change in future versions.

Finally, we define composition, $\_\circ\_ : \Gamma \models[\ q\ ]\ \Theta \to \Delta \models[\ r\ ]\ \Gamma \to \Delta \models[\ q \sqcup r\ ]\ \Theta$ by folding substitution.

4 Proving the laws

We now present a formal proof of the categorical laws, proving each lemma only once while only using structural induction. Indeed termination isn't completely trivial but is still inferred by the termination checker.

4.1 The right identity law

Let's get the easy case out of the way: the right-identity law ($xs \circ id \equiv xs$). It is easy because it doesn't depend on any other categorical equations.

The main lemma is the identity law for the substitution functor $[id] : x\ [\ id\ ] \equiv x$. To prove the successor case, we need naturality of $suc[\ q\ ]$ applied to a variable, which can be shown by simple induction over said variable: [3]

```
+-nat[]v : i [ xs + A ] ≡ suc[ q ] (i [ xs ]) A
+-nat[]v {i = zero}   {xs = xs , x} = refl
+-nat[]v {i = suc j A} {xs = xs , x} = +-nat[]v {i = j}
```

The identity law is now easily provable by structural induction:

```
[id] {x = zero}   = refl
[id] {x = suc i A} =
  i [ id + A ] ≡⟨ +-nat[]v {i = i} ⟩
  suc (i [ id ]) A
   ≡⟨ cong (λ j → suc j A) ([id] {x = i}) ⟩
  suc i A   ■
```

```
[id] {x = ` i}   =
  cong `_ ([id] {x = i})
[id] {x = t · u} =
  cong₂ _·_ ([id] {x = t}) ([id] {x = u})
[id] {x = λ t}   =
  cong λ_ ([id] {x = t})
```

Note that the $\lambda\_$ case is easy here: we need the law to hold for $t : \Gamma , A \vdash[\ T\ ] B$, but this is still covered by the inductive hypothesis because $id\ \{\Gamma = \Gamma , A\} = id \uparrow A$.

Note also that is the first time we use Agda's syntax for equational derivations. This is just syntactic sugar for constructing an equational derivation using transitivity, exploiting Agda's flexible syntax. Here $e \equiv\langle\ p\ \rangle\ e'$ means that p is a proof of $e \equiv e'$. Later we will also use the special case $e \equiv\langle\rangle\ e'$ which means that e and e' are definitionally equal (this corresponds to $e \equiv\langle\ refl\ \rangle\ e'$ and is just used to make the proof more readable). The proof is terminated with ■ which inserts refl. We also make heavy use of congruence $cong\ f : a \equiv b \to f\ a \equiv f\ b$ and a version for binary functions $cong_2\ g : a \equiv b \to c \equiv d \to g\ a\ c \equiv g\ b\ d$.

The category law $\circ id : xs \circ id \equiv xs$ is now simply a fold of the functor law ([id]).

4.2 The left identity law

We need to prove the left identity law mutually with the second functor law for substitution. This is the main lemma for associativity.

Let's state the functor law but postpone the proof until the next section: $[\circ] : x\ [\ xs \circ ys\ ] \equiv x\ [\ xs\ ]\ [\ ys\ ]$. This implicitly relies on the definitional equality[4] $\sqcup\sqcup : q \sqcup (r \sqcup s) = (q \sqcup r) \sqcup s$ because the left hand side has the type $\Delta \vdash[\ q \sqcup (r \sqcup s)\ ] A$ while the right hand side has type $\Delta \vdash[\ (q \sqcup r) \sqcup s\ ] A$.

[3] We are using the naming conventions introduced in sections 2 and 3, e.g. $i : \Gamma \ni A$.

[4] We rely on Agda's rewrite rules here. Alternatively we would have to insert a transport using subst.

Of course, we must also state the left-identity law $\mathsf{id} \circ : \mathsf{id} \circ \mathsf{xs} \equiv \mathsf{xs}$. Similarly to id, Agda will not accept a direct implementation of $\mathsf{id}\circ$ as structurally recursive. Unfortunately, adapting the law to deal with a Sort-polymorphic id complicates matters: when xs is a renaming (i.e. at sort V) composed with an identity substition (i.e. at sort T), its sort must be lifted on the RHS (e.g. by extending the $\mathsf{tm}\sqsubseteq$ functor to lists of terms) to obey $\_ \sqcup \_$.

Accounting for this lifting is certainly do-able, but in keeping with the single-responsibility principle of software design, we argue it is neater to consider only V-sorted id here and worry about equations involving Sort-coercions later (in 5.2). Therefore, we instead add a "dummy" Sort argument (i.e. $\mathsf{id}\circ' : \mathsf{Sort} \to \mathsf{id} \circ \mathsf{xs} \equiv \mathsf{xs}$) to track the size decrease (such that we can eventually just use $\mathsf{id}\circ = \mathsf{id}\circ'\ \mathsf{V}$). [5]

To prove $\mathsf{id}\circ'$, we need the β-law for $\_^+\_$, $\mathsf{xs}^+\mathsf{A} \circ (\mathsf{ys} , \mathsf{x}) \equiv \mathsf{xs} \circ \mathsf{ys}$, which can be shown with a fold over a corresponding property for $\mathsf{suc}[\_]$.

```
suc[] : (suc[ q ] x _) [ ys , y ] ≡ x [ ys ]
suc[] {q = V} = refl
suc[] {q = T} {x = x} {ys = ys} {y = y} =
  (suc[ T ] x _) [ ys , y ] ≡⟨⟩
  x [ id ^ _ ] [ ys , y ]   ≡⟨ sym ([∘] {x = x}) ⟩
  x [ (id ^ _) ∘ (ys , y) ]
   ≡⟨ cong (λ ρ → x [ ρ ]) ^∘ ⟩
  x [ id ∘ ys ]
   ≡⟨ cong (λ ρ → x [ ρ ]) id∘ ⟩
  x [ ys ] ■
```

```
^∘ : xs ^ A ∘ (ys , x) ≡ xs ∘ ys
^∘ {xs = ε}      = refl
^∘ {xs = xs , x} =
  cong₂ _,_ (^∘ {xs = xs}) (suc[] {x = x})
id∘' {xs = ε}      _ = refl
id∘' {xs = xs , x} _ = cong₂ _,_
  (id ^ _ ∘ (xs , x) ≡⟨ ^∘ {xs = id} ⟩
  id ∘ xs            ≡⟨ id∘ ⟩
  xs                 ■)
  refl
```

One may note that $^+\circ$ relies on itself indirectly via $\mathsf{suc}[]$. Like with the substitution operations, termination is justified here by the Sort decreasing.

4.3 Associativity

We finally get to the proof of the second functor law ($[\circ] : \mathsf{x}\,[\,\mathsf{xs} \circ \mathsf{ys}\,] \equiv \mathsf{x}\,[\,\mathsf{xs}\,]\,[\,\mathsf{ys}\,]$), the main lemma for associativity. The main obstacle is that for the $\lambda\_$ case; we need the second functor law for context extension: $\uparrow\circ : (\mathsf{xs} \circ \mathsf{ys}) \uparrow \mathsf{A} \equiv (\mathsf{xs} \uparrow \mathsf{A}) \circ (\mathsf{ys} \uparrow \mathsf{A})$.

To verify the variable case we also need that $\mathsf{tm}\sqsubseteq$ commutes with substitution, $\mathsf{tm}[] : \mathsf{tm}\sqsubseteq \sqsubseteq\mathsf{t}\,(\mathsf{x}\,[\,\mathsf{xs}\,]) \equiv (\mathsf{tm}\sqsubseteq \sqsubseteq\mathsf{t}\,\mathsf{x})\,[\,\mathsf{xs}\,]$, which is easy to prove by case analysis.

We are now ready to prove $[\circ]$ by structural induction:

[5] Perhaps surprisingly, this "dummy" argument does not even need to be of type Sort to satisfy Agda here. More discussion on this trick can be found at Agda issue #7693, but in summary:

- Agda considers all base constructors (constructors with no parameters) to be of minimal size structurally, so their presence can track size preservation of other base-constructor arguments across function calls.
- It turns out that a strict decrease in Sort is not necessary everywhere for termination: the context also gets structurally smaller in the call to $\_^+\_$ from id.

```
[∘] {x = zero}  {xs = xs , x}          = refl
[∘] {x = suc i _} {xs = xs , x}        =
  [∘] {x = i}
[∘] {x = ` x}   {xs = xs} {ys = ys} =
  tm⊑ ⊑t (x [ xs ∘ ys ])
    ≡⟨ cong (tm⊑ ⊑t) ([∘] {x = x}) ⟩
  tm⊑ ⊑t (x [ xs ] [ ys ])
    ≡⟨ tm[] {x = x [ xs ]} ⟩
  (tm⊑ ⊑t (x [ xs ])) [ ys ] ■

[∘] {x = t · u}                        =
  cong₂ _·_ ([∘] {x = t}) ([∘] {x = u})
[∘] {x = λ t} {xs = xs} {ys = ys} =
  cong λ_ (
    t [ (xs ∘ ys) ↑ _ ]
      ≡⟨ cong (λ zs → t [ zs ]) ↑∘ ⟩
    t [ (xs ↑ _) ∘ (ys ↑ _) ]
      ≡⟨ [∘] {x = t} ⟩
    (t [ xs ↑ _ ]) [ ys ↑ _ ] ■)
```

Associativity ∘∘ : xs ∘ (ys ∘ zs) ≡ (xs ∘ ys) ∘ zs can be proven merely by a fold of [∘] over substitutions.

However, we are not done yet. We still need to prove the second functor law for _↑_ (↑∘). It turns out that this depends on the naturality of weakening $^+$−nat∘ : xs ∘ (ys $^+$ A) ≡ (xs ∘ ys) $^+$ A, which unsurprisingly must be shown by establishing a corresponding property for substitutions: $^+$-nat[] : x [xs $^+$ A] ≡ suc[_] (x [xs]) A. The case q = V is just the naturality for variables which we have already proven ($^+$-nat[]v). The case for q = T is more interesting and relies again on [∘] and ∘id:

```
+-nat[] {q = T} {A = A} {x = x} {xs = xs} =
  x [ xs + A ]        ≡⟨ cong (λ zs → x [ zs + A ]) (sym ∘id) ⟩
  x [ (xs ∘ id) + A ] ≡⟨ cong (λ zs → x [ zs ]) (sym (+− nat∘ {xs = xs})) ⟩
  x [ xs ∘ (id + A) ] ≡⟨ [∘] {x = x} ⟩
  x [ xs ] [ id + A ] ■
```

It also turns out we need zero[] : zero[q] [xs , x] ≡ tm⊑ (⊑⊔r {q = q}) x, the β-law for zero[_], which holds definitionally in the case for either Sort.

Finally, we have all the ingredients to prove the second functor law ↑∘: [6]

```
↑∘ {r = r} {s = s} {xs = xs} {ys = ys} {A = A} =
  (xs ∘ ys) ↑ A                  ≡⟨⟩
  (xs ∘ ys) + A , zero[ r ⊔ s ]  ≡⟨ cong₂ _,_ (sym (+− nat∘ {xs = xs})) refl ⟩
  xs ∘ (ys + A) , zero[ r ⊔ s ]
    ≡⟨ cong₂ _,_ refl (tm⊑zero (⊑⊔r {r = s} {q = r})) ⟩
  xs ∘ (ys + A) , tm⊑ (⊑⊔r {q = r}) zero[ s ]
    ≡⟨ cong₂ _,_ (sym (+∘ {xs = xs})) (sym (zero[] {q = r} {x = zero[ s ]})) ⟩
  (xs + A) ∘ (ys ↑ A) , zero[ r ] [ ys ↑ A ] ≡⟨⟩
  (xs ↑ A) ∘ (ys ↑ A)            ■
```

5 Initiality

We can do more than just prove that we have a category. Indeed we can verify the laws of a simply typed category with families (CwF). CwFs are mostly known as models of dependent type theory, but they can be specialised to simple types [8]. We summarize the definition of a simply typed CwF as follows:

- A category of contexts (Con) and substitutions (_ ⊨ _),
- A set of types Ty,
- For every type A a presheaf of terms _ ⊢ A over the category of contexts (i.e. a contravariant functor into the category of sets),

[6] Actually, we also need that zero commutes with tm⊑: that is for any q⊑r : q ⊑ r we have that tm⊑zero q⊑r : zero[r] ≡ tm⊑ q⊑r zero[q].

- A terminal object (the empty context) and a context extension operation $\_ \triangleright \_$ such that $\Gamma \models \Delta \triangleright A$ is naturally isomorphic to $(\Gamma \models \Delta) \times (\Gamma \vdash A)$.

I.e. a simply typed CwF is just a CwF where the presheaf of types is constant. We will give the precise definition in the next section, hence it isn't necessary to be familiar with the categorical terminology to follow the rest of the paper.

We can add further constructors like function types $\_ \Rightarrow \_$. These usually come with a natural isomorphisms, giving rise to β and η laws, but since we are only interested in substitutions, we don't assume these. Instead we add the term formers for application $(\_ \cdot \_)$ and lambda-abstraction λ as natural transformations.

We start with a precise definition of a simply typed CwF with the additional structure to model simply typed λ-calculus (section 5.1) and then we show that the recursive definition of substitution gives rise to a simply typed CwF (section 5.2). We can define the initial CwF as a quotient inductive-inductive type (QIIT). To simplify our development, rather than using a Cubical Agda HIT[7], we postulate the existence of this QIIT in Agda (with the associated β-laws implemented with rewriting rules). By initiality, there is an evaluation functor from the initial CwF to the recursively defined CwF (defined in section 5.2). On the other hand, we can embed the recursive CwF into the initial CwF; this corresponds to the embedding of normal forms into λ-terms, only that here we talk about *substitution normal forms*. We then show that these two structure maps are inverse to each other and hence that the recursively defined CwF is indeed initial (section 5.3). The two identities correspond to completeness and stability in the language of normalisation functions.

5.1 Simply Typed CwFs

We define a record to capture simply typed CWFs, **record** CwF-simple : Set_1.

For the contents, we begin with the category of contexts, using the same naming conventions as introduced previously:

$$\begin{aligned}
&\mathsf{Con} && : \mathsf{Set} \\
&_ \models _ && : \mathsf{Con} \to \mathsf{Con} \to \mathsf{Set} \\
&\mathsf{id} && : \Gamma \models \Gamma \\
&_\circ_ && : \Delta \models \Theta \to \Gamma \models \Delta \to \Gamma \models \Theta
\end{aligned}
\qquad
\begin{aligned}
&\mathsf{id}\circ && : \mathsf{id} \circ \delta \equiv \delta \\
&\circ\mathsf{id} && : \delta \circ \mathsf{id} \equiv \delta \\
&\circ\circ && : (\xi \circ \theta) \circ \delta \equiv \xi \circ (\theta \circ \delta)
\end{aligned}$$

We introduce the set of types and associate a presheaf with each type:

$$\begin{aligned}
&\mathsf{Ty} && : \mathsf{Set} \\
&_ \vdash _ && : \mathsf{Con} \to \mathsf{Ty} \to \mathsf{Set} \\
&_[_] && : \Gamma \vdash A \to \Delta \models \Gamma \to \Delta \vdash A
\end{aligned}
\qquad
\begin{aligned}
&[\mathsf{id}] && : (\mathsf{t}\,[\,\mathsf{id}\,]) \equiv \mathsf{t} \\
&[\circ] && : \mathsf{t}\,[\,\theta\,]\,[\,\delta\,] \equiv \mathsf{t}\,[\,\theta \circ \delta\,]
\end{aligned}$$

The category of contexts has a terminal object (the empty context), and context extension resembles categorical products but mixing contexts and types:

$$\begin{aligned}
&\bullet && : \mathsf{Con} \\
&\varepsilon && : \Gamma \models \bullet \\
&_ \triangleright _ && : \mathsf{Con} \to \mathsf{Ty} \to \mathsf{Con} \\
&_,_ && : \Gamma \models \Delta \to \Gamma \vdash A \to \Gamma \models (\Delta \triangleright A) \\
&\pi_0 && : \Gamma \models (\Delta \triangleright A) \to \Gamma \models \Delta \\
&\pi_1 && : \Gamma \models (\Delta \triangleright A) \to \Gamma \vdash A
\end{aligned}
\qquad
\begin{aligned}
&\bullet\text{-}\eta && : \delta \equiv \varepsilon \\
&\triangleright\text{-}\beta_0 && : \pi_0\,(\delta\,,\mathsf{t}) \equiv \delta \\
&\triangleright\text{-}\beta_1 && : \pi_1\,(\delta\,,\mathsf{t}) \equiv \mathsf{t} \\
&\triangleright\text{-}\eta && : (\pi_0\,\delta\,,\pi_1\,\delta) \equiv \delta \\
&\pi_0\circ && : \pi_0\,(\theta \circ \delta) \equiv \pi_0\,\theta \circ \delta \\
&\pi_1\circ && : \pi_1\,(\theta \circ \delta) \equiv (\pi_1\,\theta)\,[\,\delta\,]
\end{aligned}$$

We can define the morphism part of the context extension functor as before:

[7] Cubical Agda still lacks some essential automation, e.g. integrating no-confusion properties into pattern matching.

```
_↑_ : Γ ⊨ Δ → ∀ A → Γ ▷ A ⊨ Δ ▷ A
δ ↑ A = (δ ∘ (π₀ id)) , π₁ id
```

We need to add the specific components for simply typed λ-calculus; we add the type constructors, the term constructors and the corresponding naturality laws:

```
o       : Ty                          λ_ : Γ ▷ A ⊢ B → Γ ⊢ A ⇒ B
_⇒_ : Ty → Ty → Ty                    ·[] : (t · u) [ δ ] ≡ (t [ δ ]) · (u [ δ ])
_·_ : Γ ⊢ A ⇒ B → Γ ⊢ A → Γ ⊢ B       λ[] : (λ t) [ δ ] ≡ λ (t [ δ ↑ _ ])
```

5.2 The CwF of recursive substitutions

We are building towards a proof of initiality for our recursive substitution syntax, but shall start by showing that our recursive substitution syntax obeys the specified CwF laws, specifically that CwF-simple can be instantiated with $\_ \vdash [\_]\_ / \_ \models [\_]\_$. This will be more-or-less enough to implement the "normalisation" direction of our initial CwF $\simeq$ recursive substition syntax isomorphism.

Most of the work to prove these laws was already done in section 4 but there are a couple tricky details with fitting into the exact structure the CwF-simple record requires.

Our first non-trivial decision is which type family to interpret substitutions into. In our first attempt, we tried to pair renamings/substitutions with their sorts to stay polymorphic, is-cwf .CwF.$\_ \models \_ = \Sigma$ Sort $(\Delta \models [\_]\ \Gamma)$. Unfortunately, this approach quickly breaks. The $\bullet - \eta$ CwF law forces us to provide a unique morphism to the terminal context (i.e. a unique weakening from the empty context); Σ Sort $(\Delta \models [\_]\ \Gamma)$ is simply too flexible here, allowing both V , ε and T , ε.

Therefore, we instead fix the sort to T.

```
is-cwf .CwF._ ⊨ _ = _ ⊨[ T ]_      is-cwf .CwF. • -η {δ = ε} = refl
is-cwf .CwF.▪     = •              is-cwf .CwF._∘_           = _∘_
is-cwf .CwF.ε     = ε              is-cwf .CwF. ∘∘           = sym ∘∘
```

The lack of flexibility over sorts when constructing substitutions does, however, make identity a little trickier. id doesn't fit CwF.id directly as it produces a renaming $\Gamma \models [\ \mathsf{V}\ ]\ \Gamma$. We need the equivalent substitution $\Gamma \models [\ \mathsf{T}\ ]\ \Gamma$.

We first extend tm⊑ to renamings/substitutions with a fold: tm*⊑ : q ⊑ s → Γ ⊨ [q] Δ → Γ ⊨ [s] Δ, and nd prove various lemmas about how tm*⊑ coercions can be lifted outside of our substitution operators:

```
⊑∘ : tm*⊑ v⊑t xs ∘ ys ≡ xs ∘ ys     ⊑⁺ : tm*⊑ ⊑t xs ⁺ A ≡ tm*⊑ v⊑t (xs ⁺ A)
∘⊑ : xs ∘ tm*⊑ v⊑t ys ≡ xs ∘ ys     ⊑↑ : tm*⊑ v⊑t xs ↑ A ≡ tm*⊑ v⊑t (xs ↑ A)
t[⊑] : t [ tm*⊑ v⊑t ys ] ≡ t [ ys ]  v[⊑] : i [ tm*⊑ v⊑t ys ] ≡ tm⊑ v⊑t i [ ys ]
```

Most of these are proofs come out easily by induction on terms and substitutions so we skip over them. Perhaps worth noting though is that $\sqsubseteq^+$ requires folding over substitutions using one new law, relating our two ways of weakening variables.

```
suc[id⁺] : i [ id ⁺ A ] ≡ suc i A
suc[id⁺] {i = i} {A = A} =
  i [ id ⁺ A ]      ≡⟨ ⁺-nat[]v {i = i} ⟩
  suc (i [ id ]) A  ≡⟨ cong (λ j → suc j A) [id] ⟩
  suc i A ■
```

We can now build an identity substitution by applying this coercion to the identity renaming: is-cwf .CwF.id = tm*⊑ v⊑t id. The left and right identity CwF laws take the

form $\mathsf{tm}{*}\sqsubseteq\ \mathsf{v}\sqsubseteq\mathsf{t}\ \mathsf{id}\circ\delta\ \equiv\ \delta$ and $\delta\circ\mathsf{tm}{*}\sqsubseteq\ \mathsf{v}\sqsubseteq\mathsf{t}\ \mathsf{id}\ \equiv\ \delta$. This is where we can take full advantage of the $\mathsf{tm}{*}\sqsubseteq$ machinery; these lemmas let us reuse our existing $\mathsf{id}\circ/\circ\mathsf{id}$ proofs!

```
is-cwf .CwF.id∘ {δ = δ} =
  tm*⊑ v⊑t id ∘ δ ≡⟨ ⊑∘ ⟩
  id ∘ δ          ≡⟨ id∘ ⟩
  δ ■

is-cwf .CwF.∘ id {δ = δ} =
  δ ∘ tm*⊑ v⊑t id ≡⟨ ∘⊑ ⟩
  δ ∘ id          ≡⟨ ∘id ⟩
  δ ■
```

Similarly to substitutions, we must fix the sort of our terms to T (in this case, so we can prove the identity law - note that applying the identity substitution to a variable i produces the distinct term $\grave{}\ \mathsf{i}$).

```
is-cwf .CwF.[id] {t = t} =
  t [ tm*⊑ v⊑t id ] ≡⟨ t[⊑] {t = t} ⟩
  t [ id ]          ≡⟨ [id] ⟩
  t                 ■

is-cwf .CwF._⊢_ = _⊢[ T ]_
is-cwf .CwF._[_] = _[_]
```

We now define projections $\pi_0\ (\delta\ ,\ \mathsf{t})\ =\ \delta$ and $\pi_1\ (\delta\ ,\ \mathsf{t})\ =\ \mathsf{t}$ and $\rhd{-}\beta_0$, $\rhd{-}\beta_1$, $\rhd{-}\eta$, $\pi_0\circ$ and $\pi_1\circ$ all hold by definition (at least, after matching on the guaranteed-non-empty substitution).

Finally, we can deal with the cases specific to simply typed λ-calculus. $\cdot[]$ also holds by definition, but the β-rule for substitutions applied to lambdas requires a bit of equational reasoning due to differing implementations of $\_\uparrow\_$.

```
is-cwf .CwF.λ[] {A = A} {t = x} {δ = ys} =
  λ x [ ys ↑ A ]                ≡⟨ cong (λ ρ → λ x [ ρ ↑ A ]) (sym ∘id) ⟩
  λ x [ (ys ∘ id) ↑ A ]         ≡⟨ cong (λ ρ → λ x [ ρ , ` zero ]) (sym ^−nat∘) ⟩
  λ x [ ys ∘ id ^ A , ` zero ]  ≡⟨ cong (λ ρ → λ x [ ρ , ` zero ]) (sym (∘⊑ {ys = id ^ _})) ⟩
  λ x [ ys ∘ tm*⊑ v⊑t (id ^ A) , ` zero ] ■
```

We have shown our recursive substitution syntax satisfies the CwF laws, but we want to go a step further and show initiality: that our syntax is isomorphic to the initial CwF.

An important first step is to actually define the initial CwF (and its eliminator). We use postulates and rewrite rules instead of a Cubical Agda higher inductive type (HIT) because of technical limitations mentioned previously. We also reuse our existing datatypes for contexts and types for convenience (note terms do not occur inside types in STLC).

To avoid name clashes between our existing syntax and the initial CwF constructors, we annotate every ICwF constructor with $^{\mathrm{I}}$. e.g. $\_\vdash^{\mathrm{I}}\_\ :\ \mathsf{Con}\ \to\ \mathsf{Ty}\ \to\ \mathsf{Set}$, $\mathsf{id}^{\mathrm{I}}\ :\ \Gamma\ \models^{\mathrm{I}}\ \Gamma$ etc. Note we reuse the definitions of contexts and types as in STLC there are no non-trivial equations on these components.

We state the eliminator for the initial CwF in terms of $\mathsf{Motive}\ :\ \mathrm{Set}_1$ and $\mathsf{Methods}\ :\ \mathsf{Motive}\ \to\ \mathrm{Set}_1$ records as in [4]. Again to avoid name clashes, we annotate fields of these records (corresponding to how each type/constructor is eliminated) with $^{\mathrm{M}}$.

```
module _ {𝕄} (𝐦 : Methods 𝕄) where
  elim-con : ∀ Γ → Con^M Γ      elim-cwf  : ∀ t^I → Tm^M (elim-con Γ) (elim-ty A) t^I
  elim-ty  : ∀ A → Ty^M A       elim-cwf* : ∀ δ^I → Tms^M (elim-con Δ) (elim-con Γ) δ^I
```

To state the dependent equations in $\mathsf{Methods}$ between outputs of the eliminator, enforcing congruence of equality (e.g. the functor law, which asks for $\mathsf{t}^{\mathrm{M}}\ [\ \sigma^{\mathrm{M}}\ ]^{\mathrm{M}}\ [\ \delta^{\mathrm{M}}\ ]^{\mathrm{M}}$ and $\mathsf{t}^{\mathrm{M}}\ [\ \sigma^{\mathrm{M}}\circ^{\mathrm{M}}\delta^{\mathrm{M}}\ ]^{\mathrm{M}}$ to be equated) we need dependent identity types $\_\equiv[\_]\equiv\_\ :\ \mathsf{A}\ \to\ \mathsf{A}\equiv\mathsf{B}\ \to\ \mathsf{B}\ \to\ \mathsf{Set}\ \ell$. We can define these simply by matching on the identity between A and B, $\mathsf{x}\ \equiv[\ \mathsf{refl}\ ]\equiv\ \mathsf{y}\ =\ \mathsf{x}\ \equiv\ \mathsf{y}$.

Normalisation from the initial CwF into substitution normal forms now only needs a way to connect our notion of "being a CwF" with our initial CwF's eliminator: specifically, that any set of type families satisfying the CwF laws gives rise to a Motive and associated

set of Methods. To achieve this, we define cwf-to-motive : CwF-simple → Motive and cwf-to-methods : CwF-simple → Methods, which simply project out the relevant fields, and then implement e.g. rec-cwf = elim-cwf cwf-to-methods.

The one extra ingredient we need to make this work out neatly is to introduce a new reduction for cong, cong (λ _ → x) p ≡ refl[8], via an Agda rewrite rule. This enables the no-longer-dependent _ ≡[_]≡ _s to collapse to _ ≡ _s automatically.

Normalisation into our substitution normal forms can now be achieved by with:

```
norm : Γ ⊢I A → rec-con is-cwf Γ ⊢[ T ] rec-ty is-cwf A
norm = rec-cwf is-cwf
```

Of course, normalisation shouldn't change the type of a term, or the context it is in, so we might hope for a simpler signature Γ ⊢I A → Γ ⊢[T] A and, conveniently, rewrite rules (rec-con is-cwf Γ ≡ Γ and rec-ty is-cwf A ≡ A) can get us there!

```
norm : Γ ⊢I A → Γ ⊢[ T ] A
norm = rec-cwf is-cwf

norm∗ : Δ ⊨I Γ → Δ ⊨[ T ] Γ
norm∗ = rec-cwf∗ is-cwf
```

The inverse operation to inject our syntax back into the initial CwF is easily implemented by recursion on substitution normal forms.

```
⌜_⌝  : Γ ⊢[ q ] A → Γ ⊢I A
⌜_⌝∗ : Δ ⊨[ q ] Γ → Δ ⊨I Γ
⌜ zero ⌝    = zeroI
⌜ suc i B ⌝ = sucI ⌜ i ⌝ B
⌜ ` i ⌝     = ⌜ i ⌝

⌜ t · u ⌝ = ⌜ t ⌝ ·I ⌜ u ⌝
⌜ λ t ⌝   = λI ⌜ t ⌝
⌜ ε ⌝∗    = εI
⌜ δ , x ⌝∗ = ⌜ δ ⌝∗ ,I ⌜ x ⌝
```

5.3 Proving initiality

We have implemented both directions of the isomorphism. Now to show this truly is an isomorphism and not just a pair of functions between two types, we must prove that norm and ⌜_⌝ are mutual inverses - i.e. stability (norm ⌜ t ⌝ ≡ t) and completeness (⌜ norm t ⌝ ≡ t).

We start with stability, as it is considerably easier. There are just a couple details worth mentioning:

- To deal with variables in the `_ case, we phrase the lemma in a slightly more general way, taking expressions of any sort and coercing them up to sort T on the RHS.
- The case for variables relies on a bit of coercion manipulation and our earlier lemma equating i [id ⁺ B] and suc i B.

```
stab : norm ⌜ x ⌝ ≡ tm⊑ ⊑t x
stab {x = zero}    = refl
stab {x = suc i B} =
  norm ⌜ i ⌝ [ tm∗⊑ v⊑t (id ⁺ B) ] ≡⟨ t[⊑] {t = norm ⌜ i ⌝} ⟩
  norm ⌜ i ⌝ [ id ⁺ B ]            ≡⟨ cong (λ j → suc[ _ ] j B) (stab {x = i}) ⟩
  ` i [ id ⁺ B ]                   ≡⟨ cong `_ suc[id⁺] ⟩
  ` suc i B ■
stab {x = ` i}     = stab {x = i}
stab {x = t · u}   = cong₂ _·_ (stab {x = t}) (stab {x = u})
stab {x = λ t}     = cong λ_ (stab {x = t})
```

To prove completeness, we must instead induct on the initial CwF itself, which means there are many more cases. We start with the motive:

[8] This definitional identity also holds natively in Cubical.

463 $\mathsf{compl\text{-}\mathbb{M}} : \mathsf{Motive}$

464
$$\mathsf{compl\text{-}\mathbb{M}}.\mathsf{Tm}^{\mathrm{M}}\ \_\ \_\ \mathsf{t}^{\mathrm{I}} = \ulcorner\ \mathsf{norm}\ \mathsf{t}^{\mathrm{I}}\ \urcorner \equiv \mathsf{t}^{\mathrm{I}} \qquad \mathsf{compl\text{-}\mathbb{M}}.\mathsf{Con}^{\mathrm{M}}\ \_ = \top$$
$$\mathsf{compl\text{-}\mathbb{M}}.\mathsf{Tms}^{\mathrm{M}}\ \_\ \_\ \delta^{\mathrm{I}} = \ulcorner\ \mathsf{norm}{*}\ \delta^{\mathrm{I}}\ \urcorner{*} \equiv \delta^{\mathrm{I}} \qquad \mathsf{compl\text{-}\mathbb{M}}.\mathsf{Ty}^{\mathrm{M}}\ \_ = \top$$

465 To show these identities, we need to prove that our various recursively defined syntax
466 operations are preserved by $\ulcorner\_\urcorner$.

467 Preservation of $\mathsf{zero}[\_]$, $\ulcorner\mathsf{zero}\urcorner : \ulcorner\ \mathsf{zero}[\ \mathsf{q}\ ]\ \urcorner \equiv \mathsf{zero}^{\mathrm{I}}$ reduces to reflexivity after splitting
468 on the sort.

469 Preservation of each of the projections out of sequences of terms (e.g. $\ulcorner\ \pi_0\ \delta\ \urcorner{*} \equiv$
470 $\pi_0^{\mathrm{I}}\ulcorner\ \delta\ \urcorner{*}$) reduce to the associated β-laws of the initial CwF (e.g. $\rhd{-}\beta_0^{\mathrm{I}}$).

471 Preservation proofs for $\_[\_]$, $\_ \uparrow \_$, $\_^+\_$, id and $\mathsf{suc}[\_]$ are all mutually inductive,
472 mirroring their original recursive definitions. We must stay polymorphic over sorts and again
473 use our dummy Sort argument trick when implementing $\ulcorner\mathsf{id}\urcorner$ to keep Agda's termination
474 checker happy.

475
$$\ulcorner[]\urcorner : \ulcorner\ \mathsf{x}\ [\ \mathsf{ys}\ ]\ \urcorner \equiv \ulcorner\ \mathsf{x}\ \urcorner\ [\ \ulcorner\ \mathsf{ys}\ \urcorner{*}\ ]^{\mathrm{I}} \qquad \ulcorner\mathsf{suc}\urcorner : \ulcorner\ \mathsf{suc}[\ \mathsf{q}\ ]\ \mathsf{x}\ \mathsf{B}\ \urcorner \equiv \ulcorner\ \mathsf{x}\ \urcorner\ [\ \mathsf{wk}^{\mathrm{I}}\ ]^{\mathrm{I}}$$
$$\ulcorner\uparrow\urcorner : \ulcorner\ \mathsf{xs} \uparrow \mathsf{A}\ \urcorner{*} \equiv \ulcorner\ \mathsf{xs}\ \urcorner{*} \uparrow^{\mathrm{I}} \mathsf{A} \qquad \ulcorner\mathsf{id}\urcorner' : \mathsf{Sort} \to \ulcorner\ \mathsf{id}\ \{\Gamma = \Gamma\}\ \urcorner{*} \equiv \mathsf{id}^{\mathrm{I}}$$
$$\ulcorner+\urcorner : \ulcorner\ \mathsf{xs}^{+}\ \mathsf{A}\ \urcorner{*} \equiv \ulcorner\ \mathsf{xs}\ \urcorner{*} \circ^{\mathrm{I}} \mathsf{wk}^{\mathrm{I}} \qquad \ulcorner\mathsf{id}\urcorner = \ulcorner\mathsf{id}\urcorner'\ \mathsf{V}$$
$$\ulcorner\mathsf{id}\urcorner : \ulcorner\ \mathsf{id}\ \{\Gamma = \Gamma\}\ \urcorner{*} \equiv \mathsf{id}^{\mathrm{I}}$$

476 To complete these proofs, we also need β-laws for our initial CwF substitutions, so we
477 derive these now.

478
$$\mathsf{zero}[]^{\mathrm{I}} : \mathsf{zero}^{\mathrm{I}}\ [\ \delta^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{t}^{\mathrm{I}}\ ]^{\mathrm{I}} \equiv \mathsf{t}^{\mathrm{I}}$$
$$\begin{aligned}
&\mathsf{zero}[]^{\mathrm{I}}\ \{\delta^{\mathrm{I}} = \delta^{\mathrm{I}}\}\ \{\mathsf{t}^{\mathrm{I}} = \mathsf{t}^{\mathrm{I}}\} = \\
&\quad \mathsf{zero}^{\mathrm{I}}\ [\ \delta^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{t}^{\mathrm{I}}\ ]^{\mathrm{I}} && \equiv\langle\ \mathsf{sym}\ \pi_1 \circ^{\mathrm{I}}\ \rangle \\
&\quad \pi_1^{\mathrm{I}}\ (\mathsf{id}^{\mathrm{I}} \circ^{\mathrm{I}} (\delta^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{t}^{\mathrm{I}})) && \equiv\langle\ \mathsf{cong}\ \pi_1^{\mathrm{I}}\ \mathsf{id} \circ^{\mathrm{I}}\ \rangle \\
&\quad \pi_1^{\mathrm{I}}\ (\delta^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{t}^{\mathrm{I}}) && \equiv\langle\ \rhd{-}\beta_1^{\mathrm{I}}\ \rangle \\
&\quad \mathsf{t}^{\mathrm{I}} && \blacksquare
\end{aligned}$$
$$\mathsf{suc}[]^{\mathrm{I}} : \mathsf{suc}^{\mathrm{I}}\ \mathsf{t}^{\mathrm{I}}\ \mathsf{B}\ [\ \delta^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{u}^{\mathrm{I}}\ ]^{\mathrm{I}} \equiv \mathsf{t}^{\mathrm{I}}\ [\ \delta^{\mathrm{I}}\ ]^{\mathrm{I}}$$
$$\mathsf{suc}[]^{\mathrm{I}} = \ldots$$
$$,\circ^{\mathrm{I}} : (\delta^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{t}^{\mathrm{I}}) \circ^{\mathrm{I}} \sigma^{\mathrm{I}} \equiv (\delta^{\mathrm{I}} \circ^{\mathrm{I}} \sigma^{\mathrm{I}})\ ,^{\mathrm{I}}\ (\mathsf{t}^{\mathrm{I}}\ [\ \sigma^{\mathrm{I}}\ ]^{\mathrm{I}})$$
$$,\circ^{\mathrm{I}} = \ldots$$

479 We also need a couple lemmas about how $\ulcorner\_\urcorner$ treats terms of different sorts identically:
480 $\ulcorner\sqsubseteq\urcorner : \forall\ \{\mathsf{x} : \Gamma \vdash [\ \mathsf{q}\ ]\ \mathsf{A}\} \to \ulcorner\ \mathsf{tm} \sqsubseteq \sqsubseteq \mathsf{t}\ \mathsf{x}\ \urcorner \equiv \ulcorner\ \mathsf{x}\ \urcorner$ and $\ulcorner\sqsubseteq\urcorner{*} : \ulcorner\ \mathsf{tm}{*} \sqsubseteq \sqsubseteq \mathsf{t}\ \mathsf{xs}\ \urcorner{*} \equiv \ulcorner\ \mathsf{xs}\ \urcorner{*}$.

481 We can now proceed with the preservation proofs. There are quite a few cases to cover,
482 so for brevity we elide the proofs of $\ulcorner[]\urcorner$ and $\ulcorner\mathsf{suc}\urcorner$.

483 $$\ulcorner\uparrow\urcorner\ \{\mathsf{q} = \mathsf{q}\} = \mathsf{cong}_2\ \_,^{\mathrm{I}}\_\ \ulcorner+\urcorner\ (\ulcorner\mathsf{zero}\urcorner\ \{\mathsf{q} = \mathsf{q}\})$$

484
$$\begin{aligned}
&\ulcorner+\urcorner\ \{\mathsf{xs} = \varepsilon\} && = \mathsf{sym}\ \bullet{-}\eta^{\mathrm{I}} \\
&\ulcorner+\urcorner\ \{\mathsf{xs} = \mathsf{xs}\ ,\ \mathsf{x}\}\ \{\mathsf{A} = \mathsf{A}\} = \\
&\quad \ulcorner\ \mathsf{xs}^{+}\ \mathsf{A}\ \urcorner{*}\ ,^{\mathrm{I}}\ \ulcorner\ \mathsf{suc}[\ _\ ]\ \mathsf{x}\ \mathsf{A}\ \urcorner \\
&\quad \equiv\langle\ \mathsf{cong}_2\ _,^{\mathrm{I}}_\ \ulcorner+\urcorner\ (\ulcorner\mathsf{suc}\urcorner\ \{\mathsf{x} = \mathsf{x}\})\ \rangle \\
&\quad (\ulcorner\ \mathsf{xs}\ \urcorner{*} \circ^{\mathrm{I}} \mathsf{wk}^{\mathrm{I}})\ ,^{\mathrm{I}}\ (\ulcorner\ \mathsf{x}\ \urcorner\ [\ \mathsf{wk}^{\mathrm{I}}\ ]^{\mathrm{I}}) \\
&\quad \equiv\langle\ \mathsf{sym}\ ,\circ^{\mathrm{I}}\ \rangle \\
&\quad (\ulcorner\ \mathsf{xs}\ \urcorner{*}\ ,^{\mathrm{I}}\ \ulcorner\ \mathsf{x}\ \urcorner) \circ^{\mathrm{I}} \mathsf{wk}^{\mathrm{I}}\ \blacksquare
\end{aligned}$$
$$\begin{aligned}
&\ulcorner\mathsf{id}\urcorner'\ \{\Gamma = \bullet\}\ _ && = \mathsf{sym}\ \bullet{-}\eta^{\mathrm{I}} \\
&\ulcorner\mathsf{id}\urcorner'\ \{\Gamma = \Gamma \rhd \mathsf{A}\}\ _ = \\
&\quad \ulcorner\ \mathsf{id}^{+}\ \mathsf{A}\ \urcorner{*}\ ,^{\mathrm{I}}\ \mathsf{zero}^{\mathrm{I}} && \equiv\langle\ \mathsf{cong}\ (_,^{\mathrm{I}}\ \mathsf{zero}^{\mathrm{I}})\ \ulcorner+\urcorner\ \rangle \\
&\quad \ulcorner\ \mathsf{id}\ \urcorner{*} \uparrow^{\mathrm{I}} \mathsf{A} && \equiv\langle\ \mathsf{cong}\ (_ \uparrow^{\mathrm{I}} \mathsf{A})\ \ulcorner\mathsf{id}\urcorner\ \rangle \\
&\quad \mathsf{id}^{\mathrm{I}} \uparrow^{\mathrm{I}} \mathsf{A} && \equiv\langle\ \mathsf{cong}\ (_,^{\mathrm{I}}\ \mathsf{zero}^{\mathrm{I}})\ \mathsf{id} \circ^{\mathrm{I}}\ \rangle \\
&\quad \mathsf{wk}^{\mathrm{I}}\ ,^{\mathrm{I}}\ \mathsf{zero}^{\mathrm{I}} && \equiv\langle\ \rhd{-}\eta^{\mathrm{I}}\ \rangle \\
&\quad \mathsf{id}^{\mathrm{I}} && \blacksquare
\end{aligned}$$

485 We also prove preservation of substitution composition $\ulcorner\circ\urcorner : \ulcorner\ \mathsf{xs} \circ \mathsf{ys}\ \urcorner{*} \equiv \ulcorner\ \mathsf{xs}\ \urcorner{*} \circ^{\mathrm{I}} \ulcorner\ \mathsf{ys}\ \urcorner{*}$
486 in similar fashion, folding $\ulcorner[]\urcorner$.

487 The main cases of $\mathsf{compl\text{-}}\mathbf{m} : \mathsf{Methods}\ \mathsf{compl\text{-}\mathbb{M}}$ can now be proved by just applying the
488 preservation lemmas and inductive hypotheses, e.g:

489
$$\begin{aligned}
&\mathsf{compl\text{-}}\mathbf{m}.\mathsf{id}^{\mathrm{M}} = \\
&\quad \ulcorner\ \mathsf{tm}{*} \sqsubseteq\ \mathsf{v} \sqsubseteq \mathsf{t}\ \mathsf{id}\ \urcorner{*} && \equiv\langle\ \ulcorner\sqsubseteq\urcorner{*}\ \rangle \\
&\quad \ulcorner\ \mathsf{id}\ \urcorner{*} && \equiv\langle\ \ulcorner\mathsf{id}\urcorner\ \rangle \\
&\quad \mathsf{id}^{\mathrm{I}}\ \blacksquare
\end{aligned}$$
$$\begin{aligned}
&\mathsf{compl\text{-}}\mathbf{m}._\circ^{\mathrm{M}}_\ \{\sigma^{\mathrm{I}} = \sigma^{\mathrm{I}}\}\ \{\delta^{\mathrm{I}} = \delta^{\mathrm{I}}\}\ \sigma^{\mathrm{M}}\ \delta^{\mathrm{M}} = \\
&\quad \ulcorner\ \mathsf{norm}{*}\ \sigma^{\mathrm{I}} \circ \mathsf{norm}{*}\ \delta^{\mathrm{I}}\ \urcorner{*} && \equiv\langle\ \ulcorner\circ\urcorner\ \rangle \\
&\quad \ulcorner\ \mathsf{norm}{*}\ \sigma^{\mathrm{I}}\ \urcorner{*} \circ^{\mathrm{I}} \ulcorner\ \mathsf{norm}{*}\ \delta^{\mathrm{I}}\ \urcorner{*} && \equiv\langle\ \mathsf{cong}_2\ _\circ^{\mathrm{I}}_\ \sigma^{\mathrm{M}}\ \delta^{\mathrm{M}}\ \rangle \\
&\quad \sigma^{\mathrm{I}} \circ^{\mathrm{I}} \delta^{\mathrm{I}}\ \blacksquare
\end{aligned}$$

490 The remaining cases correspond to the CwF laws, which must hold for whatever type
491 family we eliminate into in order to retain congruence of $\_ \equiv \_$. In our completeness

proof, we are eliminating into equations, and so all of these cases are higher identities (demanding we equate different proof trees for completeness, instantiated with the LHS/RHS terms/substitutions).

In a univalent type theory, we might try and carefully introduce additional coherences to our initial CwF to try and make these identities provable without the sledgehammer of set truncation (which prevents eliminating the initial CwF into any non-set).

As we are working in vanilla Agda, we'll take a simpler approach, and rely on UIP ($\mathsf{duip} : \forall \{\mathsf{p} : \mathsf{x} \equiv \mathsf{y}\}\,\{\mathsf{q} : \mathsf{z} \equiv \mathsf{w}\} \to \mathsf{p} \equiv[\,\mathsf{r}\,]\equiv \mathsf{q}$[9]), enabling e.g. $\mathsf{compl\text{-}}\mathbf{m}\ \mathsf{.id}\circ^{\mathrm{M}} = \mathsf{duip}$. And completeness is just one call to the eliminator away.

$$\mathsf{compl} : \ulcorner\, \mathsf{norm}\ \mathsf{t}^{\mathrm{I}}\ \urcorner \equiv \mathsf{t}^{\mathrm{I}}$$
$$\mathsf{compl}\ \{\mathsf{t}^{\mathrm{I}} = \mathsf{t}^{\mathrm{I}}\} = \mathsf{elim\text{-}cwf}\ \mathsf{compl\text{-}}\mathbf{m}\ \mathsf{t}^{\mathrm{I}}$$

6 Conclusions and further work

The subject of the paper is a problem which we expect everybody (including ourselves) would have thought to be trivial. As it turns out, it isn't, and we spent quite some time going down alleys that didn't work, and getting to grips with the subtleties of Agda's termination checking.

It is perhaps worth mentioning that the convenience of our solution heavily relies on Agda's built-in support for lexicographic termination [2]. This is in contrast to Rocq and Lean; the former's Fixpoint command merely supports structural recursion on a single argument and the latter has only raw elimination principles as primitive. Luckily, both of these proof assistants layer on additional commands/tactics to support more natural use of non-primitive induction, making our approach at least somewhat transferable[10].

One reviewer asked about another alternative: since we are merging $\_ \ni \_$ and $\_ \vdash \_$ why not go further and merge them entirely? Instead of a separate type for variables, one could have a term corresponding to de Bruijn index zero (written $\bullet : \Gamma \rhd \mathsf{A} \vdash' \mathsf{A}$ and an explicit weakening operator on terms (written $\_\uparrow : \Gamma \vdash' \mathsf{B} \to \Gamma \rhd \mathsf{A} \vdash' \mathsf{B}$). This has the unfortunate property that there is now more than one way to write terms that used to be identical. For instance, the terms $\bullet \uparrow\uparrow \cdot\ \bullet \uparrow \cdot\ \bullet$ and $(\bullet \uparrow \cdot\ \bullet) \uparrow \cdot\ \bullet$ are equivalent, where $\bullet \uparrow\uparrow$ corresponds to the variable with de Bruijn index two. A development along these lines is explored in [19].

This paper can also be seen as a preparation for the harder problem to implement recursive substitution for dependent types. This is harder, because here the typing of the constructors actually depends on the substitution laws. While such a Münchhausian [5] construction[11] should actually be possible in Agda, the theoretical underpinning of inductive-inductive-recursive definitions is mostly unexplored (with the exception of the proposal by [12]). However, there are potential interesting applications: strictifying substitution laws is essential to prove coherence of models of type theory in higher types, in the sense of HoTT.

Hence this paper has two aspects: it turns out that an apparently trivial problem isn't so easy after all, and it is a stepping stone to more exciting open questions. But before you can run you need to walk and we believe that the construction here can be useful to others.

[9] Note that proving this form of (dependent) UIP relies on type constructor injectivity (specifically, injectivity of $\_ \equiv \_$). We could use a weaker version taking an additional proof of $\mathsf{x} \equiv \mathsf{z}$, but this would be clunkier to use; Agda has no hope of inferring such a proof by unification.

[10] Indeed, Lean can be convinced that our substitution operations terminate after specifying measures similar to those in section 3.1, via the decreasing_by tactic.

[11] The reference is to Baron Münchhausen, who allegedly pulled himself out of a swamp by his own hair.

References

1 Andreas Abel. Parallel substitution as an operation for untyped de bruijn terms. Agda proof, 2011. URL: https://www.cse.chalmers.se/~abela/html/ParallelSubstitution.html.

2 Andreas Abel and Thorsten Altenkirch. A predicative analysis of structural recursion. *Journal of Functional Programming*, 12(1):1–41, January 2002.

3 Guillaume Allais, James Chapman, Conor McBride, and James McKinna. Type-and-scope safe programs and their proofs. In *Proceedings of the 6th ACM SIGPLAN Conference on Certified Programs and Proofs*, pages 195–207, 2017.

4 Thorsten Altenkirch and Ambrus Kaposi. Type theory in type theory using quotient inductive types. *SIGPLAN Not.*, 51(1):18–29, jan 2016. doi:10.1145/2914770.2837638.

5 Thorsten Altenkirch, Ambrus Kaposi, Artjoms Šinkarovs, and Tamás Végh. The münchhausen method in type theory. In *28th International Conference on Types for Proofs and Programs 2022*, page 10. Schloss Dagstuhl-Leibniz-Zentrum für Informatik, 2023.

6 Thorsten Altenkirch and Bernhard Reus. Monadic presentations of lambda terms using generalized inductive types. In *Computer Science Logic, 13th International Workshop, CSL '99*, pages 453–468, 1999.

7 Thosten Altenkirch, James Chapman, and Tarmo Uustalu. Monads need not be endofunctors. *Logical methods in computer science*, 11, 2015.

8 Simon Castellan, Pierre Clairambault, and Peter Dybjer. Categories with families: Unityped, simply typed, and dependently typed. *Joachim Lambek: The Interplay of Mathematics, Logic, and Linguistics*, pages 135–180, 2021.

9 Haskell Brooks Curry and Robert Feys. *Combinatory logic*, volume 1. North-Holland Amsterdam, 1958.

10 N. G de Bruijn. Lambda calculus notation with nameless dummies, a tool for automatic formula manipulation, with application to the Church-Rosser theorem. *Indagationes Mathematicae (Proceedings)*, 75(5):381–392, January 1972. URL: https://www.sciencedirect.com/science/article/pii/1385725872900340, doi:10.1016/1385-7258(72)90034-0.

11 Jason J Hickey. Formal objects in type theory using very dependent types. *Foundations of Object Oriented Languages*, 3:117–170, 1996.

12 Ambrus Kaposi. Towards quotient inductive-inductive-recursive types. In *29th International Conference on Types for Proofs and Programs TYPES 2023–Abstracts*, page 124, 2023.

13 Chantal Keller and Thorsten Altenkirch. Hereditary substitutions for simple types, formalized. In *Proceedings of the third ACM SIGPLAN workshop on Mathematically structured functional programming*, pages 3–10, 2010.

14 Conor McBride. Type-preserving renaming and substitution. *Journal of Functional Programming*, 2006.

15 Hannes Saffrich. Abstractions for multi-sorted substitutions. In *15th International Conference on Interactive Theorem Proving (ITP 2024)*. Schloss Dagstuhl–Leibniz-Zentrum für Informatik, 2024.

16 Hannes Saffrich, Peter Thiemann, and Marius Weidner. Intrinsically typed syntax, a logical relation, and the scourge of the transfer lemma. In *Proceedings of the 9th ACM SIGPLAN International Workshop on Type-Driven Development*, pages 2–15, 2024.

17 Kathrin Stark, Steven Schäfer, and Jonas Kaiser. Autosubst 2: reasoning with multi-sorted de bruijn terms and vector substitutions. In *Proceedings of the 8th ACM SIGPLAN International Conference on Certified Programs and Proofs*, pages 166–180, 2019.

18 The Agda Team. Agda documentation. https://agda.readthedocs.io, 2024. Accessed: 2024-08-26.

19 Philip Wadler. Explicit weakening. *Electronic Proceedings in Theoretical Computer Science*, 413:15–26, November 2024. Festschrift for Peter Thiemann. URL: http://arxiv.org/abs/2412.03124, doi:10.4204/EPTCS.413.2.